A Couple of Examples: Computing the Matrix Exponential and Square Root of a Matrix.

```
> restart;
  with(LinearAlgebra):
```

Example #1: Compute e^A where A is defined below...

```
> A := <<1,1,0>|<0,3,-1>|<-1,1,2>>;
```

$$A := \begin{bmatrix} 1 & 0 & -1 \\ 1 & 3 & 1 \\ 0 & -1 & 2 \end{bmatrix} \quad (1)$$

```
> Eigenvectors(A);
```

$$\begin{bmatrix} 2 \\ 2 \\ 2 \end{bmatrix}, \begin{bmatrix} -1 & 0 & 0 \\ 0 & 0 & 0 \\ 1 & 0 & 0 \end{bmatrix} \quad (2)$$

```
> Rank(IdentityMatrix(3));
  Rank(A-2*IdentityMatrix(3));
  Rank((A-2*IdentityMatrix(3))^2);
  Rank((A-2*IdentityMatrix(3))^3);
```

$$3$$
$$2$$
$$1$$
$$0 \quad (3)$$

```
> NullSpace((A-2*IdentityMatrix(3))^2);
  NullSpace((A-2*IdentityMatrix(3))^3);
```

$$\left\{ \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix} \right\}$$

$$\left\{ \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} \right\} \quad (4)$$

```
> ReducedRowEchelonForm(<<-1,0,1>|<-1,1,0>|<0,0,1>|<0,1,0>|<1,0,0>>);
```

$$\begin{bmatrix} 1 & 0 & 0 & -1 & -1 \\ 0 & 1 & 0 & 1 & 0 \\ 0 & 0 & 1 & 1 & 1 \end{bmatrix} \quad (5)$$

```
> v3 := <<0,0,1>>;
```

```
v2 := (A-2*IdentityMatrix(3)).v3;
v1 := ((A-2*IdentityMatrix(3))^2).v3;
```

$$v3 := \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

$$v2 := \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$$

$$v1 := \begin{bmatrix} 1 \\ 0 \\ -1 \end{bmatrix} \tag{6}$$

```
> P := <v1|v2|v3>;
  J := P^(-1).A.P;
```

$$P := \begin{bmatrix} 1 & -1 & 0 \\ 0 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix}$$

$$J := \begin{bmatrix} 2 & 1 & 0 \\ 0 & 2 & 1 \\ 0 & 0 & 2 \end{bmatrix} \tag{7}$$

```
> Diag := DiagonalMatrix([2,2,2]);
  N := <<0,0,0>|<1,0,0>|<0,1,0>>;

  J = Diag+N;
```

$$Diag := \begin{bmatrix} 2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

$$N := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 2 & 1 & 0 \\ 0 & 2 & 1 \\ 0 & 0 & 2 \end{bmatrix} = \begin{bmatrix} 2 & 1 & 0 \\ 0 & 2 & 1 \\ 0 & 0 & 2 \end{bmatrix} \tag{8}$$

```
> 'N' = N; 'N^2' = N^2; 'N^3' = N^3;
```

$$N = \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix}$$

$$N^2 = \begin{bmatrix} 0 & 0 & 1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

$$N^3 = \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \tag{9}$$

```
> expDiag := DiagonalMatrix([exp(2),exp(2),exp(2)]);
  expN := IdentityMatrix(3) + N + N^2/2;
```

$$expDiag := \begin{bmatrix} e^2 & 0 & 0 \\ 0 & e^2 & 0 \\ 0 & 0 & e^2 \end{bmatrix}$$

$$expN := \begin{bmatrix} 1 & 1 & \frac{1}{2} \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{bmatrix} \tag{10}$$

```
> expA := P.expDiag.expN.P^(-1);
```

$$expA := \begin{bmatrix} \frac{1}{2}e^2 & \frac{1}{2}e^2 & -\frac{1}{2}e^2 \\ e^2 & 2e^2 & e^2 \\ -\frac{1}{2}e^2 & -\frac{3}{2}e^2 & \frac{1}{2}e^2 \end{bmatrix} \tag{11}$$

```
> MatrixExponential(A);
```

$$\begin{bmatrix} \frac{1}{2}e^2 & \frac{1}{2}e^2 & -\frac{1}{2}e^2 \\ e^2 & 2e^2 & e^2 \\ -\frac{1}{2}e^2 & -\frac{3}{2}e^2 & \frac{1}{2}e^2 \end{bmatrix} \tag{12}$$

Example #2: Compute e^B where B is defined below...

```
> B := <<-1,2,0,0,0>|<1,1,0,1,1>|<-5,1,1,-2,-3>|<-1,0,0,0,-1>|<1,-2,
  0,-1,0>>;
```

(13)

$$B := \begin{bmatrix} -1 & 1 & -5 & -1 & 1 \\ 2 & 1 & 1 & 0 & -2 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 1 & -2 & 0 & -1 \\ 0 & 1 & -3 & -1 & 0 \end{bmatrix} \tag{13}$$

```
> Eigenvectors(B);
```

$$\begin{bmatrix} 1 \\ 1 \\ 1 \\ -1 \\ -1 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 1 & 0 \\ 1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 1 & 0 \end{bmatrix} \tag{14}$$

```
> 'lambda' = 1;
  Rank(IdentityMatrix(5));
  Rank(B-1*IdentityMatrix(5));
  Rank((B-1*IdentityMatrix(5))^2);
  Rank((B-1*IdentityMatrix(5))^3);

  'lambda' = -1;
  Rank(IdentityMatrix(5));
  Rank(B-(-1)*IdentityMatrix(5));
  Rank((B-(-1)*IdentityMatrix(5))^2);
  Rank((B-(-1)*IdentityMatrix(5))^3);
```

$$\lambda = 1$$
$$5$$
$$3$$
$$2$$
$$2$$
$$\lambda = -1$$
$$5$$
$$4$$
$$3$$
$$3 \tag{15}$$

```
> NullSpace(B-1*IdentityMatrix(5));
  NullSpace((B-1*IdentityMatrix(5))^2);
```

$$\left\{ \begin{bmatrix} 1 \\ 1 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ 1 \\ 0 \end{bmatrix} \right\}$$

$$\left\{\begin{bmatrix}1\\1\\0\\0\\1\end{bmatrix}, \begin{bmatrix}0\\1\\0\\1\\0\end{bmatrix}, \begin{bmatrix}-1\\2\\1\\0\\0\end{bmatrix}\right\} \tag{16}$$

```
> v2 := <<-1,2,1,0,0>>;
  v1 := (B-1*IdentityMatrix(5)).v2;
```

$$v2 := \begin{bmatrix}-1\\2\\1\\0\\0\end{bmatrix}$$

$$v1 := \begin{bmatrix}-1\\-1\\0\\0\\-1\end{bmatrix} \tag{17}$$

```
> ReducedRowEchelonForm(<<-1,-1,0,0,-1>|<1,1,0,0,1>|<0,1,0,1,0>>);
```

$$\begin{bmatrix}1 & -1 & 0\\0 & 0 & 1\\0 & 0 & 0\\0 & 0 & 0\\0 & 0 & 0\end{bmatrix} \tag{18}$$

```
> v3 := <<0,1,0,1,0>>;
```

$$v3 := \begin{bmatrix}0\\1\\0\\1\\0\end{bmatrix} \tag{19}$$

```
> NullSpace(B-(-1)*IdentityMatrix(5));
  NullSpace((B-(-1)*IdentityMatrix(5))^2);
```

$$\left\{\begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \\ 1 \end{bmatrix}\right\}$$

$$\left\{\begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}\right\} \tag{20}$$

```
> v5 := <<-1,1,0,0,0>>;
  v4 := (B-(-1)*IdentityMatrix(5)).v5;
```

$$v5 := \begin{bmatrix} -1 \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

$$v4 := \begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \\ 1 \end{bmatrix} \tag{21}$$

```
> P := <v1|v2|v3|v4|v5>;
  J := P^(-1).B.P;
```

$$P := \begin{bmatrix} -1 & -1 & 0 & 1 & -1 \\ -1 & 2 & 1 & 0 & 1 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 \\ -1 & 0 & 0 & 1 & 0 \end{bmatrix}$$

$$J := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 & 1 \\ 0 & 0 & 0 & 0 & -1 \end{bmatrix} \tag{22}$$

```
> Diag := DiagonalMatrix([1,1,1,-1,-1]);
  N := <<0,0,0,0,0>|<1,0,0,0,0>|<0,0,0,0,0>|<0,0,0,0,0>|<0,0,0,1,0>>;

  J = Diag+N;
```

$$Diag := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 0 & -1 \end{bmatrix}$$

$$N := \begin{bmatrix} 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 & 1 \\ 0 & 0 & 0 & 0 & -1 \end{bmatrix} = \begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 & 1 \\ 0 & 0 & 0 & 0 & -1 \end{bmatrix} \tag{23}$$

```
> 'N' = N; 'N^2' = N^2;
```

$$N = \begin{bmatrix} 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

$$N^2 = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{24}$$

```
> expDiag := DiagonalMatrix([exp(1),exp(1),exp(1),exp(-1),exp(-1)]);
  expN := IdentityMatrix(5) + N;
```

$$expDiag := \begin{bmatrix} e & 0 & 0 & 0 & 0 \\ 0 & e & 0 & 0 & 0 \\ 0 & 0 & e & 0 & 0 \\ 0 & 0 & 0 & e^{-1} & 0 \\ 0 & 0 & 0 & 0 & e^{-1} \end{bmatrix}$$

$$expN := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix} \qquad \textbf{(25)}$$

```
> expB := P.expDiag.expN.P^(-1);
```

$$expB := \begin{bmatrix} \frac{1}{2}e - \frac{1}{2}e^{-1} & \frac{1}{2}e - \frac{1}{2}e^{-1} & -\frac{5}{2}e + \frac{1}{2}e^{-1} & -\frac{1}{2}e + \frac{1}{2}e^{-1} & e^{-1} \\ e - e^{-1} & e & -e^{-1} & 0 & -e + e^{-1} \\ 0 & 0 & e & 0 & 0 \\ \frac{1}{2}e - \frac{3}{2}e^{-1} & \frac{1}{2}e - \frac{1}{2}e^{-1} & -\frac{1}{2}e - \frac{1}{2}e^{-1} & \frac{1}{2}e + \frac{1}{2}e^{-1} & -e + 2e^{-1} \\ \frac{1}{2}e - \frac{3}{2}e^{-1} & \frac{1}{2}e - \frac{1}{2}e^{-1} & -\frac{3}{2}e - \frac{1}{2}e^{-1} & -\frac{1}{2}e + \frac{1}{2}e^{-1} & 2e^{-1} \end{bmatrix} \qquad \textbf{(26)}$$

```
> MatrixExponential(B);
```

$$\left[\left[\frac{1}{2}\left(-1+e^2\right)e^{-1}, \frac{1}{2}\left(-1+e^2\right)e^{-1}, -\frac{1}{2}\left(5e^2-1\right)e^{-1}, -\frac{1}{2}\left(-1+e^2\right)e^{-1}, e^{-1}\right],\right. \qquad \textbf{(27)}$$

$$\left[\left(-1+e^2\right)e^{-1}, e, -e^{-1}, 0, -\left(-1+e^2\right)e^{-1}\right],$$

$$\left[0, 0, e, 0, 0\right],$$

$$\left[\frac{1}{2}\left(e^2-3\right)e^{-1}, \frac{1}{2}\left(-1+e^2\right)e^{-1}, -\frac{1}{2}\left(1+e^2\right)e^{-1}, \frac{1}{2}\left(1+e^2\right)e^{-1}, -\left(e^2\right.\right.$$

$$\left.\left. -2\right)e^{-1}\right],$$

$$\left.\left[\frac{1}{2}\left(e^2-3\right)e^{-1}, \frac{1}{2}\left(-1+e^2\right)e^{-1}, -\frac{1}{2}\left(3e^2+1\right)e^{-1}, -\frac{1}{2}\left(-1+e^2\right)e^{-1}, 2e^{-1}\right]\right]$$

Example #3: Compute $\sqrt{C}$ where C is defined below...

```
> C := <<-28,22,-15,22>|<-42,31,-15,27>|<-16,8,1,8>|<-32,22,-15,26>>;
```

$$C := \begin{bmatrix} -28 & -42 & -16 & -32 \\ 22 & 31 & 8 & 22 \\ -15 & -15 & 1 & -15 \\ 22 & 27 & 8 & 26 \end{bmatrix} \tag{28}$$

```
> evec := Eigenvectors(C);
```

$$evec := \begin{bmatrix} 1 \\ 9 \\ 4 \\ 16 \end{bmatrix}, \begin{bmatrix} -2 & -2 & -1 & -\frac{3}{2} \\ 1 & 1 & 0 & 1 \\ -1 & 0 & 0 & -\frac{1}{2} \\ 1 & 1 & 1 & 1 \end{bmatrix} \tag{29}$$

```
> P := evec[2];
```

$$P := \begin{bmatrix} -2 & -2 & -1 & -\frac{3}{2} \\ 1 & 1 & 0 & 1 \\ -1 & 0 & 0 & -\frac{1}{2} \\ 1 & 1 & 1 & 1 \end{bmatrix} \tag{30}$$

```
> J := DiagonalMatrix([1,9,4,16]);
```

$$J := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 9 & 0 & 0 \\ 0 & 0 & 4 & 0 \\ 0 & 0 & 0 & 16 \end{bmatrix} \tag{31}$$

```
> sqrtJ := DiagonalMatrix([1,3,2,4]);
```

$$sqrtJ := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 3 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 4 \end{bmatrix} \tag{32}$$

```
> sqrtC := P.sqrtJ.P^(-1);
```

$$sqrtC := \begin{bmatrix} -4 & -8 & -4 & -6 \\ 4 & 7 & 2 & 4 \\ -3 & -3 & 1 & -3 \\ 4 & 5 & 2 & 6 \end{bmatrix} \tag{33}$$

```
> sqrtC^2=C;
```

$$\begin{bmatrix} -28 & -42 & -16 & -32 \\ 22 & 31 & 8 & 22 \\ -15 & -15 & 1 & -15 \\ 22 & 27 & 8 & 26 \end{bmatrix} = \begin{bmatrix} -28 & -42 & -16 & -32 \\ 22 & 31 & 8 & 22 \\ -15 & -15 & 1 & -15 \\ 22 & 27 & 8 & 26 \end{bmatrix} \tag{34}$$

```
> sqrtC = MatrixFunction(C,sqrt(x),x);
```

$$\begin{bmatrix} -4 & -8 & -4 & -6 \\ 4 & 7 & 2 & 4 \\ -3 & -3 & 1 & -3 \\ 4 & 5 & 2 & 6 \end{bmatrix} = \begin{bmatrix} -4 & -8 & -4 & -6 \\ 4 & 7 & 2 & 4 \\ -3 & -3 & 1 & -3 \\ 4 & 5 & 2 & 6 \end{bmatrix} \tag{35}$$